

# 実働荷重に対する疲労き裂進展挙動の シミュレーション

石 川 浩\*

## 1. はじめに

機械・構造物が実用中に受ける実働荷重は、一般には、その上下限が複雑に変動したランダム荷重である場合が多く、このようなランダム負荷を受ける構造部材の破壊はほとんどの場合が疲労に基因したものであることは周知の事実である。したがって、ランダム荷重に対する部材の疲労寿命の予測が必須不可欠の肝要事となる。このためには実用状態で予測されるランダムに変動した荷重を実際に同じ形状・寸法の部材に与えた、いわゆる実寸実働荷重疲労試験を行ってランダム疲労寿命を求めるのが最も望ましいと考えられるが、設備や労力等の面で多額の費用を要することは不可避的である。

そこで本研究においては、近年の破壊力学の著しい進歩を踏まえて、ランダム荷重に対する疲労き裂の進展モデルを遅延（リターデーション）効果をも勘案しながら構築し、定振幅荷重に対する実験パラメータを用いてランダム荷重に対する寿命予測を実施することの可能なコンピュータ・プログラムの開発を意図した。同時に、入力情報としてのランダム荷重の統計的性質ならびに材料パラメータのばらつきを勘案した予測寿命の信頼性解析へのアプローチに関連した話題について論究した。

## 2. 疲労き裂進展に関する破壊力学的技術現況の概観

### 2.1 き裂進展法則

一般に疲労き裂の進展挙動は

---

\* 香川大学商業短期大学部 〒760 高松市幸町2-1

$$\frac{da}{dN} = f(a, \sigma_{\max}, R, T, E, \dots) \quad (2-1)$$

と表すことができよう。ここに、

$da/dN$  = き裂進展速度

$a$  = き裂長

$\sigma_{\max}$  = 最大応力

$R = \sigma_{\min}/\sigma_{\max}$  = 応力比

$T$  = 温度

$E$  = 環境効果

式(2-1)を基にして、き裂進展速度に関する様々な研究が試みられ、現在以下に述べるような種々のき裂進展法則が提案されるに至っている。

(a) Paris のパワー法則<sup>(1)</sup>

破壊力学の進歩に伴って、き裂先端の応力拡大係数  $K$  を用いて結果をうまく整理することができるが見出され、Paris は Al 合金板のデータから実験的に 4 乗則を提案した。

$$\frac{da}{dN} = f(K) = C(\Delta K)^4 \quad (2-2)$$

ここに、 $\Delta K = K_{\max} - K_{\min} = (1-R)K_{\max}$  = 応力拡大係数変動幅。

現在 Paris の法則は次式のように一般化されて用いられている。

$$\frac{da}{dN} = C(\Delta K)^n \quad (2-3)$$

式(2-3)の両辺の対数をとると

$$\ln\left(\frac{da}{dN}\right) = \ln C + n \ln(\Delta K) \quad (2-4)$$

すなわち  $da/dN$  と  $\Delta K$  は、図1に示すように、両対数グラフ上で直線関係を有することがわかる。なお、指数  $n$  はその直線の勾配を表し、また  $C$  は  $\Delta K=1$  に対するその直線の切片に関連したものである。

(b) Forman らの式<sup>(2)</sup>

$K_{\max}$  の値が破壊靱性  $K_{Ic}$  (あるいはこれに近いある値  $K_{cr}$ ) の値に近づくと、き裂進展速度が Paris の線形関係 (両対数グラフ上での) よりはずれて




図1 き裂進展速度 ( $da/dN$ ) と応力拡大係数変動幅 ( $\Delta K$ ) との関係

急増することを勘案して次式が提案された。

$$\frac{da}{dN} = \frac{C(\Delta K)^n}{(1-R)K_{IC} - \Delta K} = \frac{C(\Delta K)^n}{(1-R)(K_{IC} - K_{max})} \quad (2-5)$$

なお、定数  $C$  と指数  $n$  は Paris 則のものと同様ではあるが、必ずしも同一ではない。また、 $R$  は応力比を表すものである。

(c) Collipriest の式<sup>(3)</sup>

その後の  $\frac{da}{dN} - \Delta K$  関係の研究の進歩から、破壊靱性  $K_{IC}$  と応力拡大係数変動幅の下限値 (threshold stress intensity factor range)  $\Delta K_{TH}$  との間で  $\ln\left(\frac{da}{dN}\right) - \ln(\Delta K)$  の関係が図2に示すように S 字形を描くことが見出し、これを表現する式として次式が提案された。

$$\begin{aligned} \frac{da}{dN} = & \exp\left[n\left(\frac{\ln K_{IC} - \ln \Delta K_{TH}}{2}\right)\right] \\ & \times \tanh^{-1}\left(\frac{\ln \Delta K - \frac{\ln\{(1-R)K_{IC}\} + \ln \Delta K_{TH}}{2}}{\frac{\ln\{(1-R)K_{IC}\} - \ln \Delta K_{TH}}{2}}\right) \\ & + \ln\left\{C \exp\left(\frac{\ln K_{IC} + \ln \Delta K_{TH}}{2} \times n\right)\right\} \end{aligned} \quad (2-6)$$




図2 S字形の  $\frac{da}{dN}$ - $\Delta K$  関係

Collipriest の上式はまた “Sigmoidal equation” とも呼ばれている。

(d) Walker の式<sup>(4)</sup>

定振幅疲労試験データの蓄積とともに、疲労き裂の進展速度は応力比  $R = \sigma_{\min}/\sigma_{\max}$  の影響を受けることが明らかとなり、この観点から Walker は有効応力幅の考えを導入した。すなわち、応力拡大係数を用いて表示すれば、

$$\Delta K_{\text{eff}} = (1-R)^m K_{\text{max}} \quad (2-7)$$

と表される。式(2-7)を式(2-3)の一般化された Paris 則に適用すれば、

$$\begin{aligned} \frac{da}{dN} &= C(\Delta K_{\text{eff}})^n \\ &= C[(1-R)^m K_{\text{max}}]^n \end{aligned} \quad (2-8)$$

となり、さらに上式を

$$\begin{aligned} \Delta K &= K_{\text{max}} - K_{\text{min}} \\ &= (1-R)K_{\text{max}} \end{aligned} \quad (2-9)$$

を用いて  $\Delta K$  表示すれば、最終的には

$$\frac{da}{dN} = C \left[ \frac{\Delta K}{(1-R)^{1-m}} \right]^n \quad (2-10)$$

という Walker のき裂進展速度の式を得る。なお、上式において  $m=1$  とす

れば

$$\frac{da}{dN} = C(\Delta K)^n$$

となり、これは式 (2-3) の Paris の一般法則に一致する。

(e) Hall らの式<sup>(5)</sup>

Hall らは米空軍 (USAF) のき裂進展データに適合させるための実験式として次式を提案した。

$$\frac{da}{dN} = C(K_{max} - K_{TH})^\alpha (\Delta K)^n \quad (2-11)$$

ここに、 $C$ ,  $\alpha$ ,  $n$  は curve-fitting から得られる材料定数である。式 (2-11) はまた “Boeing equation” とも呼ばれている。

(f) Bell らの式<sup>(6)</sup>

変動荷重負荷下においては降伏点を超える過荷重 (overload) がしばしば作用することがあり、このような過荷重を受けると、き裂の進展は遅延 (retardation) せしめられる現象が生じる。このような遅延効果 (retardation effect) を勘案したものとして Bell らは次式を提案した。

$$\frac{da}{dN} = C[(1+q\bar{R})\Delta K]^n \quad (2-12)$$

ここに  $\bar{R}$  は、

$$\begin{cases} \bar{R} = R & \text{for } R \leq R_{cut} \\ \bar{R} = R_{cut} & \text{for } R > R_{cut} \end{cases} \quad (2-13)$$

ただし、

$R_{cut}$  = 応力比の上限打切り値

で定義される修正応力比であり、またパラメータ  $C$ ,  $q$ ,  $n$  は curve-fitting から得られる材料実験定数である。

なお、上式は Elber の closure model<sup>(7)</sup> である次式

$$\begin{cases} \Delta K_{eff} = U\Delta K = \left( \frac{\sigma_{max} - \sigma_{op}}{\sigma_{max} - \sigma_{min}} \right) \Delta K \\ U = 0.5 + 0.4R \quad (-0.1 < R < 0.7\dots) \end{cases} \quad (2-14)$$

2024 T3 アルミニウム合金の場合)

$$\left\{ \begin{array}{l} R = \sigma_{\min} / \sigma_{\max} \\ \sigma_{op} = \text{き裂開口応力値 (crack opening stress)} \end{array} \right.$$

を修正したものであって、実験データにうまく適合するように考えられている。

## 2.2 リターデーション・モデル

先述した overload によるき裂進展速度の遅延(リターデーション)効果を勘案したモデルは現在いくつか提案されているが、本研究では Willenborg モデルと Vroman モデルを取扱うこととした。

### (a) Willenborg モデル<sup>(9)</sup>

Wheeler によって提案されたモデル<sup>(8)</sup>ではわずらわしいデータの当てはめ作業を行う必要があったが、それを省略する目的から生み出されたのが Willenborg モデルである。このモデルでは以下の諸仮定が採用されている。

- (i) リターデーション(遅延)は各々の荷重サイクルにおける最大応力とそれ以前に受けた overload との関数である。
- (ii) リターデーションは overload によって生じた残留応力によって以後の最大応力が減少せしめられる量に比例する。
- (iii) リターデーション効果は overload によって生じたき裂先端の降伏域の長さによって漸減・減衰する。
- (iv) 以前に受けた overload より大きな overload を新たに受ける場合には、すべての過去の条件とは全く独立に、新しいリターデーションの条件式が生成される。
- (v) 負荷応力はすべて非負である。

さて、図3に示すように、き裂長  $a_{o1}$  の時点で overload  $\sigma_{o1}$  の負荷が完了したものとすれば、この overload によるき裂先端の塑性域の大きさ  $R_{y_{o1}}$  は Irwin によれば次式で与えられる。

$$R_{y_{o1}} = \frac{1}{c\pi} \left( \frac{K_{\max o1}}{\sigma_y} \right)^2 \quad (2-15)$$




図3 き裂先端の降伏域

ここに、

$$\left. \begin{aligned}
 K_{\max o1} &= \text{overload } \sigma_{o1} \text{ による応力拡大係数 (最大値)} \\
 \sigma_y &= \text{材料の降伏応力} \\
 c &= 2 \text{ (平面応力条件)} \\
 &= 4\sqrt{2} \text{ (平面ひずみ条件)}
 \end{aligned} \right\} \quad (2-16)$$

したがって、overload  $\sigma_{o1}$  によって影響を受ける部分は

$$a_{po1} = a_{o1} + R_{yol} \quad (2-17)$$

となる。

き裂長  $a = a_c (< a_{po1})$  における作用応力を  $\sigma_c$  とし、その時の応力拡大係数を  $K_{\max}$  とすれば、 $\sigma = \sigma_c$  でのき裂先端の塑性域の大きさは、

$$R_{yc} = \frac{1}{c\pi} \left( \frac{K_{\max}}{\sigma_y} \right)^2 \quad (2-18)$$

したがって、 $\sigma_c$  によって影響を受ける領域の大きさは、

$$a_{pc} = a_c + R_{yc} \tag{2-19}$$

式(2-17) および (2-19) によってリターデーションの条件式が作り出される。

すなわち、

- (i) もし、 $a_{pc} > a_{poi}$  であれば仮定から新しいリターデーションの条件式が作られ、

$$\begin{cases} a_{poi} \leftarrow a_{pc} \\ R_{yoi} \leftarrow R_{yc} \\ K_{maxoi} \leftarrow K_{max} \end{cases}$$

と置き換えられることになる。

- (ii) もし、 $a_{pc} \leq a_{poi}$  であれば、き裂進展は遅延せしめられる。またこの遅延効果は以下のようにして計算される 残留応力によってもたらされる。すなわち現時点で残存している塑性領域

$$R_{yap} = a_{poi} - a_c \tag{2-20}$$

と同等の塑性領域を生み出す応力  $\sigma_{ap}$  に関連した応力拡大係数  $K_{maxap}$ 、換言すれば、次式を満足する  $K_{maxap}$

$$R_{yap} = a_{poi} - a_c = \frac{1}{c\pi} \left( \frac{K_{maxap}}{\sigma_y} \right)^2 \tag{2-21}$$

を求め、これによって得られる  $\sigma_{ap}$  を用いて、残留応力

$$\sigma_{res} = \sigma_{ap} - (\sigma_c)_{max} \tag{2-22}$$

したがってまた応力拡大係数

$$K_{res} = K_{maxap} - K_{max} \tag{2-23}$$

を計算する。この残留応力による遅延は、

$$\left. \begin{aligned} (K_{max})' &= K_{max} - K_{res} \\ &= 2K_{max} - K_{maxap} \\ (K_{min})' &= K_{min} - K_{res} \\ &= (K_{max} + K_{min}) - K_{maxap} \\ R' &= (K_{min})' / (K_{max})' \end{aligned} \right\} \tag{2-24}$$

として新しく  $(K_{max})'$ 、 $(K_{min})'$  および  $R'$  を計算し直すことによって

評価することができる。

式(2-21)ならびに(2-23)の関係は図4に示す通りである。なお、式(2-15)、(2-17)ならびに(2-21)から




図4 リターデーションによる残留応力

$$(K_{\max ap})^2 = c\pi\sigma_y^2(a_{pol} - a_c)$$

それゆえ、

$$\begin{aligned} K_{\max ap} &= K_{\max ol} \sqrt{\frac{a_{pol} - a_c}{R_{yol}}} \\ &= K_{\max ol} \sqrt{\frac{a_{pol} - a_c}{a_{pol} - a_{ol}}} \end{aligned} \tag{2-25}$$

$$a_c = a_{pol} - (a_{pol} - a_{ol}) \left( \frac{K_{\max ap}}{K_{\max ol}} \right)^2 \tag{2-26}$$

Willenborg のモデルによるリターデーション効果は式(2-24)で評価できるわけであるが、このモデルでは仮定(V)から非負の応力のみを扱うので、負の応力はこれを0と置き換える必要がある。また overload  $\sigma_{o1}$  の負荷直後では  $a_c = a_{ol}$  であるから式(2-25)より  $K_{\max ap} = K_{\max ol}$  である。したがって、overload  $\sigma_{o1}$  を受けた後、その後引き続き受ける応力  $\sigma_c$  の大小関係によって、次の3つのモードのリターデーション効果が識別されることにな

る。

- (i) モード I— $\Delta K$  と  $R$  の減少に基づくリターデーション・モード  
このモードは  $(K_{\max})' > 0$ ,  $(K_{\min})' \leq 0$  の場合に対応する。

したがって、

$$\left. \begin{aligned} K_{\min} + K_{\max} &\leq K_{\max o1} < 2K_{\max} \\ (\sigma_c)_{\min} + (\sigma_c)_{\max} &\leq \sigma_{o1} < 2(\sigma_c)_{\max} \end{aligned} \right\} \quad (2-27-a)$$

この場合には  $(K_{\max})_{\text{eff}} = (K_{\max})'$ ,  $(K_{\min})_{\text{eff}} = 0$  となるから、

$$\left. \begin{aligned} \Delta K_{\text{eff}} &= (K_{\max})' = K_{\max} - K_{\text{res}} \\ R_{\text{eff}} &= (K_{\min})_{\text{eff}} / (K_{\max})_{\text{eff}} = 0 \end{aligned} \right\} \quad (2-27-b)$$

- (ii) モード II— $R$  のみの減少に基づくリターデーション・モード  
これは  $(K_{\min})' > 0$  の場合に対応するものである。

したがって、

$$\left. \begin{aligned} K_{\min} + K_{\max} &> K_{\max o1} \\ (\sigma_c)_{\min} + (\sigma_c)_{\max} &> \sigma_{o1} \end{aligned} \right\} \quad (2-28-a)$$

この場合には  $(K_{\max})_{\text{eff}} = (K_{\max})'$ ,  $(K_{\min})_{\text{eff}} = (K_{\min})'$  となるから、

$$\left. \begin{aligned} \Delta K_{\text{eff}} &= (K_{\max})_{\text{eff}} - (K_{\min})_{\text{eff}} \\ &= K_{\max} - K_{\min} = \Delta K \\ R_{\text{eff}} &= (K_{\min})_{\text{eff}} / (K_{\max})_{\text{eff}} \\ &= (K_{\min} - K_{\text{res}}) / (K_{\max} - K_{\text{res}}) \end{aligned} \right\} \quad (2-28-b)$$

- (iii) モード III—最大のリターデーションを得るモード

このモードではき裂成長は完全に停止し、 $(K_{\max})' \leq 0$  の場合に対応する。

したがって、

$$\left. \begin{aligned} 2K_{\max} &\leq K_{\max o1} \\ 2(\sigma_c)_{\max} &\leq \sigma_{o1} \end{aligned} \right\} \quad (2-29-a)$$

この場合には  $(K_{\max})_{\text{eff}} = (K_{\min})_{\text{eff}} = 0$  であるから

$$\left. \begin{aligned} \Delta K_{\text{eff}} &= 0 \\ R_{\text{eff}} &= 0 \end{aligned} \right\} \quad (2-29-b)$$

以上のモード I ~ III に対応した負荷応力状態を模式的に図 5 に示す。なお、 $\sigma_{o1}/\sigma_c$  を overload ratio (過荷重比) と称するが、Willenborg のモデルではこの overload ratio が 2 以上でき裂進展が停止することが予測される。事実、アルミ合金では、 $\sigma_{o1}/\sigma_c \geq 2.3$  で進展の停止することが報告されている。<sup>(10)</sup>




図 5 Willenborg のリターデーション・モデル

以上のように求められる  $\Delta K_{eff}$  および  $R_{eff}$  の値を第 2.1 節のいずれかのき裂進展法則—例えば、Paris 則や Walker の式に適用してき裂進展を予測することができる。

(b) Vroman モデル<sup>(11)</sup>

このモデルは (a) の Willenborg モデルと類似したものではあるが、 $\Delta K_{eff}$  および  $R_{eff}$  の算出法が異なる。すなわち、 $R_{eff}$  としては応力比  $R = \sigma_{min}/\sigma_{max}$  をそのまま用い、 $\Delta K_{eff}$  としては次式を用いるものである。

$$\left. \begin{aligned} R_{eff} &= R = K_{min}/K_{max} \text{ for } R \leq R_{cut} \\ &= R_{cut} \text{ for } R > R_{cut} \end{aligned} \right\} \quad (2-30)$$

$$\Delta K_{eff} = (K_{max} - K_{min}) - \frac{1}{3} \left[ K_{max o1} \sqrt{\frac{a_{o1} + R_{y o1} - a_c}{R_{y o1}}} - K_{max} \right] \quad (2-31)$$

ここに、

$$R_{y o1} = \frac{1}{\pi} \left( \frac{K_{max o1}}{\sigma_y} \right)^2$$

$\sigma_y$  = 材料の降伏応力

$K_{max}$  = 現荷重サイクルでの応力拡大係数の最大値

$K_{min}$  = 現荷重サイクルでの応力拡大係数の最小値

$$\left. \begin{array}{l} a_{o1} \\ R_{y o1} \\ K_{max o1} \end{array} \right\} \text{ 以前に受けた過荷重 } \sigma_{o1} \text{ による値}$$

以上で定義される  $\Delta K_{eff}$  および  $R_{eff}$  を適当なき裂進展法則に適用して用いる。例えば、Walker の式に適用した場合には

$$\frac{da}{dN} = C \left[ \frac{\Delta K_{eff}}{(1 - R_{eff})^{1-m}} \right]^n \quad (2-32)$$

ここに、 $C$ ,  $n$ ,  $m$  は材料定数であって、いずれも定振幅試験結果から求められるものである。

### 2.3 応力拡大係数の計算式

種々の試験片形状ならびに応力条件に対して詳細な計算がなされているが、<sup>(12)(13)</sup> 本研究では図6に示すように中央に貫通き裂のある帯板の一様引張りならびに片方のみに edge crack をもつ場合を取扱うこととした。




図6 中央部のき裂と端部のき裂

(i) 中央なき裂 (center crack) のある帯板の一様引張り

$$\left. \begin{aligned}
 K &= \sigma \sqrt{\pi a} F\left(\frac{2a}{w}\right) \\
 \xi &= 2a/w \\
 F(\xi) &\cong \sqrt{\sec\left(\frac{\pi\xi}{2}\right)} \quad \left( \begin{array}{l} \xi \leq 0.7 \text{ で } 0.3\% \text{ の誤差} \\ \xi = 0.8 \text{ で } 1\% \text{ の誤差} \end{array} \right)
 \end{aligned} \right\} \quad (2-33)$$

(ii) edge crack をもつ帯板の一樣引張り

$$\left. \begin{aligned}
 K &= \sigma \sqrt{\pi a} F\left(\frac{a}{w}\right) \\
 \xi &= \frac{a}{w} \\
 F(\xi) &\cong 1.12 - 0.231\xi + 10.55\xi^2 - 21.72\xi^3 + 30.39\xi^4 \\
 &\quad (\xi \leq 0.6 \text{ で誤差 } 0.5\%)
 \end{aligned} \right\} \quad (2-34)$$

### 3. リターデーションを考慮したランダム荷重に対する疲労き裂進展挙動のシミュレーション・プログラムの開発

#### 3.1 適用き裂進展法則

第 2.1 節に述べたようにき裂進展法則に関してはこれまでに種々の方法が提案されているが、最も基本的でポピュラーなものとしては、

- (i) Paris の式
- (ii) Forman の式、および
- (iii) Walker の式

を挙げることができよう。Paris の式は簡便さの故もあってよく用いられており、また Forman の式は  $K_{Ic}$  の影響を取入れている特徴的なものと思われるが、研究者の好みとしては Walker の式の方がより好んで用いられているようである。

そこで、本研究におけるコンピュータ・プログラムの開発に際しては、

- (i) Paris の式

$$\frac{da}{dN} = C(\Delta K)^n \quad (3-1)$$

(ii) Walker の式

$$\frac{da}{dN} = C \left[ \frac{\Delta K}{(1-R)^{1-m}} \right]^n \quad (3-2)$$

の2者をコントロール・パラメータ NTYPE を用いて適宜選択使用可能とした。すなわち、

NTYPE=0 → Paris の式

NTYPE≠0 → Walker の式

となるように設定することとした。

### 3.2 適用リターデーション・モデル

Willenborg モデルならびに Vroman モデルをコントロール・パラメータ NRETRD によって選択できるようにした。もちろんリターデーションを考えに入れない場合をも含めしめるよう配慮した。すなわち、

NRETRD ≤ 0 → リターデーションを考えない場合

NRETRD = 1 → Willenborg モデル

NRETRD > 0 (≠1) → Vroman モデル

リターデーションを考える場合き裂先端の塑性域は式 (2-15), すなわち

$$R_{y,01} = \frac{1}{c\pi} \left( \frac{K_{max,01}}{\sigma_y} \right)^2 \quad (3-3)$$

によって計算する必要があるが、上式の係数  $c$  は応力条件によって値が異なる。そこでこれをコントロールパラメータ NPLANE によって制御する。すなわち、

NPLANE = 0 → 平面応力条件:  $c=2$

NPLANE ≠ 0 → 平面ひずみ条件:  $c=4\sqrt{2}$

となるようにコンピュータ・プログラムを作成した。

### 3.3 き裂の形状ならびに位置

試験片形状および荷重条件によって様々なき裂形状となることが考えられるが、本研究ではいずれも貫通き裂 (through the thickness crack) を取扱う

こととし、また、き裂部位はパラメータ NLOC によって制御した。すなわち、

NLOC  $\geq 0$   $\rightarrow$  試験片中央部のき裂 (center crack)

NLOC  $< 0$   $\rightarrow$  試験片幅方向の一端から生じたき裂 (single edge crack)

### 3.4 入力ランダム荷重データ

入力ランダム荷重は1飛行 (flight) 単位で取扱うこととし、最小応力  $\sigma_{\min}$  および最大応力  $\sigma_{\max}$  のペアで荷重1サイクルと考える。荷重単位は絶対単位  $\text{kg}/\text{mm}^2$  で与えてもよいし、また設計制限応力 (design limit stress)  $\sigma_{\text{lim}}$  ( $\text{kg}/\text{mm}^2$ ) の%表示で与えてもよい。後者の場合にはプログラム内部で自動的に絶対単位に換算されるようにした。換言すれば、例えば  $\sigma'$  が  $\sigma_{\text{lim}}$  の%単位で与えられている場合、

$$\sigma = \sigma' \cdot \sigma_{\text{lim}} / 100.0 (\text{kg}/\text{mm}^2) \quad (3-4)$$

の変換を行わしめた。なお、この荷重単位はパラメータ MSCON によって制御した。すなわち、

MSCON  $\geq 0$   $\rightarrow$  絶対単位 ( $\text{kg}/\text{mm}^2$ )

MSCON  $< 0$   $\rightarrow$  % $\sigma_{\text{lim}}$  単位 (単位は%)

なお、本プログラムでは負の応力は取扱わないので、絶対単位  $\text{kg}/\text{mm}^2$  に変換された最大応力  $\sigma_{\max}$  および最小応力  $\sigma_{\min}$  が次の場合にはそれぞれの値を置き換えて用いることとした。この場合、原応力データを“AS PRESCRIBED”，また変換したデータを“AS MODIFIED”と表示することとした。

$\sigma_{\min} < 0$   $\rightarrow$   $\sigma_{\min} = 0$  ( $\text{kg}/\text{mm}^2$ )

$\sigma_{\max} < 0.01$   $\rightarrow$   $\sigma_{\max} = 0.01$  ( $\text{kg}/\text{mm}^2$ )

さらに、design limit stress  $\sigma_{\text{lim}}$  については、現在のき裂長を  $a$  とした場合に、このき裂長をもつ試験片が一様に  $\sigma_{\text{lim}}$  の負荷を受けるものとして計算した design limit stress intensity factor

$$K_{\text{lim}} = \sigma_{\text{lim}} \cdot F(a) \quad (3-5)$$

$F(a)$  = 応力拡大係数計算のための関数

が破壊靱性  $K_{\text{IC}}$  を超える場合には計算を中止させる機能をもたせた。なお、

航空機設計等によく用いられるこの design limit stress を考えに入れない場合にはパラメータ SIGLIM の値を

$$\text{SIGLIM}=0.0$$

と設定すればよい。また、入力データが1フライト中のランダム荷重履歴データであるから、その POWER 乗の平均を求め統計的性質を算定できるようにした。POWER=2.0 の場合には応力履歴の2乗平均値 (rms 値) が計算される。

### 3.5 き裂進展の積分計算式

Paris の式にしる、Walker の式にしる、それらは微分方程式の形で与えられているわけであるから、ランダム入力に応じてその微分方程式を積分することによってき裂成長が求められることになる。積分は微小区間に分けた区分求積によって実施することとなるが、この際には積分のための微小値のとり方いかんによっては計算は非常に多時間を要するものとなり、破断寿命の長い場合には著しく経済性を損ねたものとなることは容易に推察される。それゆえ、本プログラムでは以下の方式に依ることとした。

現在き裂長が  $a$  であり、荷重サイクル  $\sigma_{\min}-\sigma_{\max}$  を  $n$  回受けるものとしよう。なお、 $n=1$  のときは cycle-by-cycle の積分となる。

- (i) 負荷応力  $\sigma_{\min}$ ,  $\sigma_{\max}$  を用いて、選択した条件の下で  $\Delta K$  と  $R$ , もしくは  $\Delta K_{\text{eff}}$  と  $R_{\text{eff}}$ , を計算し、き裂進展則から  $da/dN$  を計算する。この際、き裂長の微小変化によっては進展速度ならびに retardation に大きな影響がないという実験結果に基づいて、き裂長の微小変化量として  $\Delta a=0.01a$  を考え、上の計算は平均的な値として

$$a=a+\frac{1}{2}\Delta a=1.005a$$

において行う。

- (ii)  $\Delta a/\left(\frac{da}{dN}\right)$  を与えられたサイクル数  $n$  と比較する。もしも、

(イ)  $\Delta a/(da/dN)>n$  の場合

このステップでのき裂成長は

$$(\Delta a)' = n \cdot (da/dN)$$

であると考え、き裂長を新しく  $a = a + (\Delta a)'$  として、次の荷重ステップへ進む。

(ロ)  $\Delta a / (da/dN) \leq n$  の場合

$$\left. \begin{array}{l} a = a + 0.01a \\ n = n - 0.01a / (da/dN) \end{array} \right\} \text{として、再びこの荷重ステップを}$$

(i) からの手順に基づいて再計算する。

(iii) 1フライト中のすべての荷重履歴を終了すれば、その時のき裂長を記憶しておき、再び初めの荷重ステップに立返って次のフライトに対するものとして同様の計算を繰り返す。

(iv) 破壊の判定は design limit load に対する  $K_{lim}$  もしくは maximum spectrum load に対する  $K_{max}$  のいずれか大きな方が  $K_{IC}$  を超えたときに、破壊と判定する。なお  $\Delta K_{TH}$  の効果も調べるため、予め与えた  $\Delta K_{TH}$  と  $\Delta K_{off}$  を比べ、もしも、 $\Delta K_{off} \leq \Delta K_{TH}$  であればき裂進展は全くないものとして次の荷重サイクルに進むようにした。したがって、 $\Delta K_{TH} = 0.0$  を入力すれば、threshold stress intensity は考えないことと同等となる。

### 3.6 入力データとその FORMAT

本プログラムの計算に必要な入力データのサンプルを図7に示す。

図7の1行のデータがカード入力の場合の1枚のカードに対応する。以下簡単な説明を加えることとする。変数の後の( )はデータのFORMATを表すものである。

(i) 1枚目のカード(図7の第1行目)

IDENT(40A2)=計算の識別のためのID。80字まで与えられるようにした。

(ii) 2枚目のカード(図7の第2行目)

```

INPUT DATA FORMAT SAMPLE CHART -- PROBLEM IDENTIFICATION UP TO 80 CHARACTERS
SIGMAY CKIC DELKTH EXPM EXPN CONST(E12.5)RCUT
HALFW T NLOCHRETRD NTYPEPLANE NCASE
C0(I),I=1,NCASE --- FORMAT(10F8.5)
NLOADPOWER SIGLIN MCON
SGMIN(I),SGMAX(I),I=1,NLOAD --- FORMAT(10F8.3) --- MIN. AND MAX. STRESSES.
    
```

----- SAMPLE INPUT DATA FORMAT -----

```

CASE-2: WALKER'S EQUATION; WILLENBORG'S MODEL; CENTER CRACK; C0=3.0MM
50.0 60.0 2.5 0.31000 3.400 1.7500 -090.500
10.0 2.0 1 1 1 1 1
3.0
502.0 20.0 1
0.00 4.00 1.60 7.60 2.80 6.00 4.60 7.40 2.80 7.40
2.00 7.60 2.00 5.60 1.40 6.20 2.80 6.60 3.40 7.00
1.20 5.80 3.00 7.20 2.00 5.40 3.00 7.60 8.40 6.20
3.60 5.20 2.00 5.40 4.00 7.20 2.80 6.00 .80 10.60
2.60 5.40 2.80 7.20 3.40 7.80 1.80 10.40 3.80 4.80
1.80 6.80 4.80 8.00 2.00 4.00 2.60 3.80 1.80 4.60
2.60 9.20 3.60 6.20 2.20 4.60 2.20 3.60 1.60 7.80
5.00 8.60 3.00 7.40 4.40 7.60 4.60 7.60 2.80 5.00
3.00 6.40 1.80 7.40 5.80 11.80 2.60 5.80 2.20 5.60
2.80 8.60 3.00 4.80 1.60 6.60 2.20 7.60 1.80 6.60
    
```

図7 入力データの FORMAT

SIGMAY(F10.5) = 材料の降伏応力  $\sigma_y$  (kg/mm<sup>2</sup>)

CKIC(F10.5) = 破壊靱性  $K_{Ic}$  ((kg/mm<sup>2</sup>)√mm)

DELKTH(F10.5) = スレスホールド応力拡大係数  $\Delta K_{TH}$  ((kg/mm<sup>2</sup>)√mm)

EXPM(F10.5) = Walker の式における指数  $m$  の値。Paris の式を用いる場合にはこの値はダミーとなる。

EXPN(F10.5) = Walker または Paris の式における指数  $n$  の値。

CONST(E12.5) = Walker または Paris の式における定数  $C$  の値。

RCUT(F10.5) = 応力比  $R$  の上限打ち切り値  $R_{cut}$  (cut-off stress)

ratio)

(iii) 3枚目のカード (図7の第3行目)

HALFW(F10.5) = 試験片幅の半分  $\frac{w}{2}$  (mm)T(F10.5) = 試験片厚さ  $t$  (mm)

NLOC(I6) = き裂部位の判別用パラメータ

NLOC  $\geq 0$   $\rightarrow$  中央部 (center crack)NLOC  $< 0$   $\rightarrow$  端部 (single edge crack)

NRETRD(I6) = リターデーション・モデル選択パラメータ

NRETRD  $\leq 0$   $\rightarrow$  リターデーション無視NRETRD = 1  $\rightarrow$  Willenborg モデルNRETRD  $> 0$  ( $\neq 1$ )  $\rightarrow$  Vroman モデル

NTYPE(I6) = き裂進展法則の選択パラメータ

NTYPE = 0  $\rightarrow$  Paris の式NTYPE  $\neq 0$   $\rightarrow$  Walker の式

NPLANE(I6) = き裂先端の塑性域計算のための応力条件パラメータ

NPLANE = 0  $\rightarrow$  平面応力条件NPLANE  $\neq 0$   $\rightarrow$  平面ひずみ条件

NCASE(I6) = 同一応力条件の下で (同一荷重の下で) 種々の初期き裂長に対する寿命を予測する場合の異なる初期き裂長の数

(iv) 4枚目のカード (図7の第4行目)

C0(I), I=1, NCASE...異なる初期き裂長 (10F8.5) で NCASE 個まで入れることができる。NCASE が 10 より大きな場合には同じ FORMAT で 5枚目のカードにも入ることになる。

(v) 5枚目のカード (図7の第5行目)

$NLOAD(I6) = 1$  フライト中での荷重サイクルの数

$POWER(F10.5) =$  荷重の累乗平均を求めるための指数。もしも  
 $POWER = 2$  ならば, rms 応力値となる。

$SIGLIM(F10.5) =$  design limit stress  $\sigma_{lim}$  ( $kg/mm^2$ )。これを  
 考えない場合は  $SIGLIM = 0.0$  とする。

$MSCON(I6) = 1$  フライト中での荷重履歴の値の単位の判別パ  
 ラメータ

$MSCON \geq 0 \rightarrow kg/mm^2$  単位

$MSCON < 0 \rightarrow$  制限応力  $\sigma_{lim}$  の%単位

(vi) 6枚目のカード (図7の第6行目) 以降

荷重履歴が  $\sigma_{min}$ ,  $\sigma_{max}$  のペアで  $NLOAD$  個入る (10F8.3)。1枚の  
 カードには  $\sigma_{min}$ ,  $\sigma_{max}$  のペアで5組 (5つの荷重サイクル) が入るよ  
 うになっているので, 相当数のカードが並ぶことになる。

#### 4. プログラムの計算例と考察

実際の実験データがあればシミュレーション結果の比較ができて興味深い  
 が, ここでは図7下半分に示したような仮想的なデータ (一応アルミ合金を想  
 定した) を用いて本プログラムによる計算を行った。なお, 本コンピュータ・  
 プログラムのリストならびに結果の打出し例は本論文巻末に APPENDIX と  
 して添付してあるので参照されたい。計算に当たって採用したパラメータ値は  
 以下の通りである。

$$\sigma_y = 50 \text{ kg/mm}^2$$

$$K_{IC} = 68 \text{ (kg/mm}^2\text{)}\sqrt{\text{mm}}$$

$$\Delta K_{TH} = 2.5 \text{ (kg/mm}^2\text{)}\sqrt{\text{mm}}$$

$$m = 0.310$$

$$n = 3.40$$

$$C = 1.750 \times 10^{-9}$$

$$R_{cut} = 0.5$$

$$w/2=10.0 \text{ mm}$$

$$t=2.0 \text{ mm}$$

$$\sigma_{lim}=20 \text{ kg/mm}^2$$

入力ランダム荷重履歴としては、飛行中に受ける突風荷重を模擬し、図8に示すような荷重履歴を採用した。1フライト中の荷重サイクルは $100/2=50$ サイクルとした。




図8 突風ランダム荷重履歴の例

$\sigma_{lim}=20 \text{ kg/mm}^2$  に対して、荷重履歴の最大は約60%程度とした。

図9は Walker の式を用いて、Vroman のリターデーション・モデルを採用した場合のシミュレーション結果である。初期き裂長は、

$$a_0=1.5, 1.8, 2.0, 2.5 \text{ および } 3.0 \text{ mm}$$

の5種類とした。 $\sigma_{lim}=20.0 \text{ kg/mm}^2$  と設定した関係で、き裂長が3.2mm程度になれば  $K_{lim} > K_{IC}$  となって破壊条件を満足するので、き裂成長はこの程度で終了してしまうことがわかる。




図9 き裂進展寿命のシミュレーション結果

次に、design limit stress  $\sigma_{lim}$  の効果を見るために、 $\sigma_{lim} = 0.0 \text{ kg/mm}^2$  と設定して初期き裂長  $\alpha_0 = 3.0 \text{ mm}$  の場合をシミュレートすれば、この場合には図10に示すように、き裂は約6mm以上に成長して破壊に至ることが観察される。これによって本プログラムが満足に作動していることが明らかである。

図11には Paris と Walker のき裂進展法則によるシミュレーション結果の相違を示した。実験定数  $n$ ,  $C$  は両式で全く同一と設定したのでこのように両式の差異が大きくなったのであろうと思われる。

図12はリターデーション・モデルのシミュレーション結果に及ぼす影響について調べたものである。リターデーションを考えない場合には当然のことながら寿命は短かく安全側の結果を示すが、これに反して Willenborg のモデルは危険側の結果をもたらす。それゆえ、Vroman のモデルが妥当かと思われるが、検証すべき実際のデータがないので現段階では確かな結論付けがしがたい。




図10 design limit stress  $\sigma_{lim}$  の効果




図11 Paris の式と Walker の式の比較




図12 リターデーション・モデルの比較

以上の計算例を縦覧して明らかなように、本プログラムの妥当性と有用性が大いに示唆される。

### 5. ランダム荷重の統計的取扱いと信頼性設計

ランダム荷重負荷下における疲労寿命の信頼性解析手法は非常に興味ある課題である。この広範な体系を詳述するには少なからぬスペースを要するので、ここではこれに関する別稿の拙小論<sup>(14)~(26)</sup>を参照していただくことによって本節に代えることとしたい。

### 6. 今後の研究に対する一考案

本研究によって、ランダム荷重下での疲労寿命推定のシミュレーション・プログラムは確立されたわけであるが、これに付随して、今後の在りうるべき研究の方向について考えてみたい。

(1) 航空機等の複雑な構造物においては、存在もしくは発生する疲労き裂

の形状は種々様々であり、負荷条件も多種多様を極めるであろう。それゆえ、現実的な様々な場合に対応して、本プログラムの拡張整備を図ることが必要である。

具体的には貫通形ではないき裂形状、あるいはリベット穴などから進展するき裂等に対する推定式を充実せしめることが必要である。

## (2) き裂進展法則

$$da/dN = f(\Delta K, R, \dots)$$

を積分することによってき裂進展曲線ならびに疲労寿命を得るわけであるが、積分過程の計算には非常に長時間を要し、計算費用がばかにならないほど大となることが多いであろう。それゆえ、例えばこれを多項式近似によって直接積分の可能な数式に変換することによって、シミュレーションの実行時間を大幅に短縮する手法を研究することが実用上重要である。これによって、多種類のケースのシミュレーションを短時間に行うことができ、またモンテカルロ法を用いた信頼性解析が可能となる。

(3) 突風荷重などは定常確率過程としての取扱いが可能であって、比較的簡単にシミュレーションを実施できるが、操舵荷重等はその性質からいっても非定常性を有した確率過程と考えられる。したがって、このような非定常確率過程としてのランダム荷重のシミュレーション手法の確立を図ることが、現実的観点から非常に重要である。<sup>(27)</sup>

(4) 材料は繰返し応力の負荷を受けることによって疲労し、き裂の発生・成長を経て遂には破壊する。それゆえ、き裂の成長に伴って残留強度も減じるであろうが、このような残留強度の確率構造を明らかとし、ランダムな負荷とのからみで構造物信頼性の解析を行う手法を開発することが大切であると考えられる。

## 参 考 文 献

- (1) Paris, P. C. and Edogan, F., "A Critical Analysis of Crack Propagation Laws", J. Basic Eng., Trans. ASME, Vol. 85 (1963), pp. 528-533.
- (2) Forman, R. G., Keamey, V. E., and Engle R. M., "Numerical Analysis

- of Crack Propagation in Cyclic-Loaded Structures," Journal of Basic Engineering, Trans. of ASME, Vol. 89D(1967), p. 459.
- (8) Collipriest, J. E., Jr., and Ehret, R. M., "Computer Modeling of Part-Through-Crack Growth," SD 72-CE-15, Rockwell International, Space Division, 1972.
- (4) Walker, K., "The Effect of Stress Ratio During Crack Propagation and Fatigue for 2024-T3 and 7075-T6 Aluminum," ASTM STP 462, 1970.
- (5) Hall, L. R., Shah, R. C., and Engstrom, W. L., "Fracture and Fatigue Crack Growth Behavior of Surface Flaws and Flaws Originating at Fastener Holes," AFFDL-TR-74-47, Air Force Flight Dynamics Laboratory, WPAFB, Ohio, 1974.
- (6) Bell, P. D. and Creager, M. "Crack Growth Analysis for Arbitrary Spectrum Loading," AFFDL-TR-74-129, Air Force Flight Dynamics Laboratory, WPAFB, Ohio, 1974.
- (7) Elber, W., "The Significance of Fatigue Crack Closure," Damage Tolerance in Aircraft Structures, ASTM STP 486, American Society for Testing and Materials, 1971.
- (8) Wheeler, O. E., "Crack Propagation under Spectrum Loading," FZM-5602, June 1970, General Dynamics, Fort Worth Division.
- (9) Willenborg, J., Engle, R. M. and Wood, H. A., "A Crack Growth Retardation Model Using an Effective Stress Concept," AFFDL-TM-71 IFBR, Wright-Patterson Air Force Base, Ohio, 1971.
- (10) Probst, E. P. and Hillberry, B. M., "Fatigue Crack Delay and Arrest Due to Single Peak Tensile Overloads," AIAA Paper No. 73-325, Presented at AIAA Dynamic Specialists Conference, March 19-20, 1973, Williamsburg, Virginia U. S. A.
- (11) Vroman, G., "Crack Propagation Analysis with Retardation," presented at ASTM E-24 Committee Meeting, Northrop Corporation, Hawthorne, California, 1971.
- (12) 岡村弘之, 「線形破壊力学入門」, (1976-5), p.105, 培風館.
- (13) 石田 誠, 「き裂の弾性解析と応力拡大係数」, (1976-6), p.160, 培風館.
- (14) 石川 浩, 「実働荷重下における機械構造物疲労寿命の信頼性解析 I. 信頼性工学の基礎概念」, 材料, 24-259 (1975-4), pp.344~355.
- (15) 石川 浩, 「同上II. ワイブル破壊確率モデルならびに順序統計と TTFF」, 材料, 24-260 (1975-5), pp.477~488.
- (16) 石川 浩, 「同上III. 母数推定の統計的手法と安全寿命の決定(その1)」, 材料,

24-261 (1975-6), pp. 563~573.

- (17) 石川 浩, 「同上Ⅳ, 母数推定の統計的手法と安全寿命の決定(その2)」, 材料, 24-262 (1975-7), pp. 693~703.
- (18) 石川 浩, 「同上Ⅴ, 母数推定の統計的手法と安全寿命の決定(その3)」, 材料, 24-263 (1975-8), pp. 783~788.
- (19) 石川 浩, 「同上Ⅵ, 実働荷重の統計的性質(パワースペクトル密度分布と荷重ひん度分布)」, 材料, 24-264 (1975-9), pp. 914~924.
- (20) 石川 浩, 「同上Ⅶ, 荷重と強度の分布を考慮した信頼性設計(その1)」, 材料, 24-265 (1975-10), pp. 993-1000.
- (21) 石川 浩, 「同上Ⅷ, 荷重と強度の分布を考慮した信頼性設計(その2)」, 材料, 24-266 (1975-11), pp. 1072~1081.
- (22) 谷本敏夫, 網島貞男, 石川 浩, 「各種 FRP の片持ち回転曲げ疲労寿命の信頼性と設計安全寿命の研究」, 材料, 28-304 (1979-1), pp. 40~46.
- (23) 石川 浩, 木村 等, 「不規則現象の数学的記述とランダム疲労寿命の推定について」, 香川大学経済論叢, 52-1・2 (1979-6), pp. 96~183.
- (24) 石川 浩, 木村 等, 「強度設計と信頼性」, 香川大学経済論叢, 53-3 (1981-1), pp. 305~355.
- (25) 石川 浩, 谷本敏夫, 木村 等, M. Shinozuka, 「ワイブル寿命分布に従う機械・構造物の信頼度の推定手法について」, 材料, 30-323 (1981-1), pp. 25~31.
- (26) 石川 浩, 「信頼性設計の基礎概念」, 日本材料学会中四国支部講習会「信頼性設計の基礎と応用」教材, (1981-11), pp. 1~22.
- (27) Shinozuka, M., H. Ishikawa and H. Mitsuma, "Data-Based Nonstationary Random Processes," Proc. of the Specialty Conference on Probabilistic Mechanics and Structural Reliability, Tucson, Arizona, USA, (1979-1), pp. 39-43, ASCE.

## APPENDIX-1 ランダム荷重に対する疲労き裂進展寿命予測プログラム・リスト

```

FILE NAME: SCGPP0 ---- BY DR. H. ISHIKAWA ---- DATE: 1981-11-11

1: C.... ----- PROGRAM CGPFO -----
2: C.... THIS PROGRAM PREDICTS FATIGUE CRACK GROWTH,
3: C.... UNDER CONSTANT AND/OR VARIABLE AMPLITUDE LOADING,
4: C.... FOR THROUGH THE THICKNESS CENTER AND/OR EDGE CRACK,
5: C.... WITH CRACK GROWTH RATE EQUATION OF PARIS' AND/OR WALKER'S,
6: C.... INCLUDING RETARDATION MODEL OF WILLENBORG'S AND/OR VROMAN'S.
7: C
8:     DIMENSION IDENT(40),CO(10),SGMIN(2000),SGMAX(2000),CRACK(100)
9:     COMMON /KLOC/ CPL,HALFW,W,ROOTC,COEFK,PI2
10: C
11: C.... TO ASSIGN INPUT/OUTPUT PERIPHERAL.
12:     LIN=5
13:     LOUT=2
14: C
15: C.... TO READ IN NECESSARY INPUT INFORMATION.
16: C.... FOR PROBLEM IDENTIFICATION UP TO 80 CHARACTERS.
17:     READ(LIN,500) IDENT
18:     500 FORMAT(40A2)
19: C.... FOR MATERIAL CONSTANT.
20: C     SIGMAY= MATERIAL YIELD STRENGTH (KG/MM**2)
21: C     CKIC = FRACTURE TOUGHNESS (KG/MM**2)*SQRT(MM)
22: C     DELKTH= THRESHOLD STRESS INTENSITY (KG/MM**2)*SQRT(MM)
23: C     EXPM = EXPONENT (SMALL M) IN WALKER'S EQUATION AS
24: C           DC/DN=CONST*(DELTAK*(1.0-R)**(1.0-EXPM))**EXPN
25: C           FOR PARIS' MODEL, EXPM IS SET 1.0, PRODUCING
26: C           DC/DN=CONST*(DELTAK)**EXPN
27: C     EXPN = POWER EXPONENT FOR LN(DC/DN) VS LN(DELTAK) CURVE
28: C     CONST = INTERCEPT AT DELTAK=1.0
29: C     RCUT = CUT-OFF STRESS RATIO
30:     READ(LIN,505) SIGMAY,CKIC,DELKTH,EXPM,EXPN,CONST,RCUT
31:     505 FORMAT(SF10.5,E12.5,F10.5)
32: C.... FOR SPECIMEN GEOMETRY AND CONTROL PARAMETER.
33: C     HALFW ---- HALF WIDTH OF THE SPECIMEN (MM)
34: C           WIDTH, THEREFORE, IS W=2.0*HALFW (MM).
35: C     T ---- SPECIMEN THICKNESS (MM)
36: C     IF NLOC.GE.0 ---- CENTER CRACK
37: C     NLOC.LT.0 ---- SINGLE EDGE CRACK
38: C     IF NRETRD.GT.0 --- RETARDATION IS TAKEN INTO ACCOUNT.
39: C           NRETRD=1 --- WILLENBORG'S RETARDATION MODEL
40: C           O.W. --- VROMAN'S RETARDATION MODEL
41: C     NRETRD.LE.0 --- NO RETARDATION IS CONSIDERED.
42: C     IF NTYPE=0 --- PARIS' CRACK GROWTH RATE EQUATION
43: C     NTYPE.NE.0 --- WALKER'S CRACK GROWTH RATE EQUATION
44: C     IF NPLANE=0 --- PLANE STRESS CONDITION (C=2.0*PI)
45: C     O.W. --- PLANE STRAIN CONDITION (C=4*SQRT(2)*PI)
46: C     NCASE ---- NO. OF DIFFERENT INITIAL CRACK SIZES.
47: C           IN THIS PROGRAM NCASE IS LIMITED UP TO 10.
48: C           HOWEVER, THIS CAN BE CHANGED ARBITRARILY
49: C           IN CONJUNCTION WITH DIMENSION CHANGE IN CO(I)..
50:     READ(LIN,510) HALFW,T,NLOC,NRETRD,NTYPE,NPLANE,NCASE
51:     510 FORMAT(2F10.5,5I6)
52:     IF(NCASE.LE.0) NCASE=1
53: C.... FOR DIFFERENT INITIAL CRACK SIZES UNDER THE SAME LOADING.
54:     READ(LIN,515) (CO(I),I=1,NCASE)
55:     515 FORMAT(10F8.5)
56: C.... FOR SPECTRUM LOADING INFORMATION
57: C     NLOAD --- NO. OF LOADS IN A FLIGHT.
58: C     POWER --- POWER EXPONENT FOR STRESS AVERIZATION.
59: C     SIGLIM --- DESIGN LIMIT STRESS (KG/MM**2)
60: C     MSCON --- STRESS UNIT CONTROL PARAMETER.
61: C           MSCON.GE.0 --- IN TERMS OF KG/MM**2
62: C           O.W. --- IN TERMS OF % SIGLIM, I.E.,

```

```

63: C          SGMAX(I)=SIGLIM*SGMAX(I)/100.
64: C          SGMIN(I)=SIGLIM*SGMIN(I)/100.
65: C          STRESS SEQUENCE MUST BE IN ORDER OF MIN. FOLLOWED BY MAX.
66: READ(LIN,520) NLOAD,POWER,SIGLIM,MSCON
67: 520 FORMAT(16,2F10.5,16)
68: IF(NLOAD.LE.0) NLOAD=1
69: IF(MSCON.LT.0 .AND. SIGLIM.LE.0.0) GO TO 960
70: READ(LIN,525) (SGMIN(I),SGMAX(I),I=1,NLOAD)
71: 525 FORMAT(10F8.3)
72: C..... TO MODIFY GIVEN STRESSES IN ABSOLUTE UNIT IF NECESSARY.
73: IF(MSCON.GE.0) GO TO 70
74: SUNIT=SIGLIM/100.0
75: DO 75 I=1,NLOAD
76: SGMAX(I)=SGMAX(I)*SUNIT
77: SGMIN(I)=SGMIN(I)*SUNIT
78: 75 CONTINUE
79: C
80: C.... TO PRINT OUT INPUT INFORMATION
81: 70 IF(NTYPE.EQ.0) WRITE(LOUT,600) IDENT
82: IF(NTYPE.NE.0) WRITE(LOUT,605) IDENT
83: 600 FORMAT(1H1//1H,'PREDICTION OF CRACK GROWTH ',
84: *'BASED ON PARIS'' EQUATION'/1H,40A2)
85: 605 FORMAT(1H1//1H,'PREDICTION OF CRACK GROWTH ',
86: *'BASED ON WALKER''S EQUATION'/1H,40A2)
87: IF(NLOC.LT.0) GO TO 1
88: WRITE(LOUT,610)
89: 610 FORMAT(3X,'FOR THROUGH THE THICKNESS CENTER CRACK')
90: GO TO 2
91: 1 WRITE(LOUT,615)
92: 615 FORMAT(3X,'FOR THROUGH THE THICKNESS SINGLE EDGE CRACK')
93: 2 IF(NRETRD.LE.0) GO TO 3
94: IF(NRETRD.EQ.1) GO TO 4
95: WRITE(LOUT,620)
96: 620 FORMAT(3X,'VROMAN''S RETARDATION MODEL IS APPLIED.//')
97: GO TO 5
98: 4 WRITE(LOUT,625)
99: 625 FORMAT(3X,'WILLENBORG''S RETARDATION MODEL IS APPLIED.//')
100: GO TO 5
101: 3 WRITE(LOUT,630)
102: 630 FORMAT(3X,'NO RETARDATION IS CONSIDERED.//')
103: 5 WRITE(LOUT,635) SIGMAY,CKIC,DELKTH,EXPM,EXP,N,CONST,RCUT
104: 635 FORMAT(/1H,'MATERIAL CONSTANTS USED:/'
105: *3X,'YIELD STRENGTH SY= ',E14.7,' (KG/MM**2)/'
106: *3X,'CRITICAL S.I.F. KIC= ',E14.7,' (KG/MM**2)SQRT(MM)/'
107: *3X,'THRESHOLD S.I.F. KTH= ',E14.7,' (KG/MM**2)SQRT(MM)/'
108: *3X,'EXPONENT M M= ',E14.7/'
109: *3X,'EXPONENT N N= ',E14.7/'
110: *3X,'INTERCEPT C C= ',E14.7/'
111: *3X,'CUT-OFF RATIO RCUT= ',E14.7//')
112: WRITE(LOUT,640) HALF,T
113: 640 FORMAT(1H,'SPECIMEN GEOMETRY:/'
114: *3X,'HALF WIDTH W/2= ',E14.7,' (MM)/'
115: *3X,'THICKNESS T= ',E14.7,' (MM)/'
116: WRITE(LOUT,645) SIGLIM
117: 645 FORMAT(///1X,'INPUT SPECTRUM LOADING INFORMATION/'
118: *3X,'DESIGN LIMIT STRESS SIGLIM =',E13.6,' (KG/MM**2)')
119: WRITE(LOUT,650)
120: WRITE(LOUT,652)(SGMAX(I),I=1,NLOAD)
121: WRITE(LOUT,654)
122: WRITE(LOUT,652)(SGMIN(I),I=1,NLOAD)
123: 650 FORMAT(//1H,'MAXIMUM STRESS AS PRESCRIBED (KG/MM**2)')
124: 652 FORMAT(1X,10(F7.2,1X))
125: 654 FORMAT(/1H,'MINIMUM STRESS AS PRESCRIBED (KG/MM**2)')
126: C
127: C.... TO SET UP PARAMETERS FOR COMPUTATION.
128: PI=3.1415926
129: P12=PI/2.0
130: W=2.0*HALFW

```

```

131:         IF(RCUT.GE.1.0) RCUT=0.99
132:         IF(POWER.LE.0.0) POWER=1.0
133:         FNLOAD=FLOAT(NLOAD)
134:         RPOWER=1.0/POWER
135:         PLANE=2.0*PI
136:         IF(NPLANE.NE.0) PLANE=4.0*SQR(2.0)*PI
137:         CTR=HALFW
138:         IF(NLOC.LT.0) CTR=W
139:         NNLOC=NLOC
140:         EXPM1=EXPM-1.0
141: C.... FOR PARIS' CRACK GROWTH RATE EQUATION, SET EXPM=1.0
142:         IF(NTYPE.EQ.0) EXPM1=0.0
143: C
144: C.... TO GET STATISTICAL PROPERTIES OF SPECTRUM LOADING
145: C     IF MAX. STRESS IS LESS THAN 0.01 KG/MM**2, SGMAX(I)=0.01
146: C     IF MIN. STRESS IS LESS THAN 0.0 KG/MM**2, SGMIN(I)=0.0
147:         SGM1=0.0
148:         SGM2=0.0
149:         SR1=0.0
150:         SR2=0.0
151:         DO 60 I=1,NLOAD
152:         SGM1=SGM1+(SGMAX(I)-SGMIN(I))*POWER
153:         IF(SGMAX(I).LT.0.01) SGMAX(I)=0.01
154:         IF(SGMIN(I).LT.0.0) SGMIN(I)=0.0
155:         SGM2=SGM2+(SGMAX(I)-SGMIN(I))*POWER
156:         RATIO=SGMIN(I)/SGMAX(I)
157:         SR1=SR1+RATIO
158:         RATIO=ABS(RATIO)
159:         SR2=SR2+RATIO**POWER
160:     60 CONTINUE
161:         SGM1=(SGM1/FNLOAD)**RPOWER
162:         SGM2=(SGM2/FNLOAD)**RPOWER
163:         SR1=SR1/FNLOAD
164:         SR2=(SR2/FNLOAD)**RPOWER
165:         WRITE(LOUT,656)
166:         WRITE(LOUT,658) (SGMAX(I),I=1,NLOAD)
167:         WRITE(LOUT,660)
168:         WRITE(LOUT,658) (SGMIN(I),I=1,NLOAD)
169:     656 FORMAT(/1H , 'MAXIMUM STRESS AS MODIFIED (KG/MM**2)')
170:     658 FORMAT(1X,10(F7.2,1X))
171:     660 FORMAT(/1H , 'MINIMUM STRESS AS MODIFIED (KG/MM**2)')
172:         WRITE(LOUT,662) SGM1
173:         WRITE(LOUT,664) SGM2
174:         WRITE(LOUT,666) SR1,SR2
175:     662 FORMAT(/1H , 'PRESCRIBED MEAN STRESS RANGE S1=',E12.5)
176:     664 FORMAT(1H , 'MODIFIED MEAN STRESS RANGE S2=',E12.5)
177:     666 FORMAT(/1H 'SIMPLE MEAN STRESS-RATIO R1=',F8.4/
178:         *1H , 'RAISED MEAN STRESS-RATIO R2=',F8.4/)
179: C
180: C.... TO COMPUTE FOR EACH INITIAL CRACK SIZE.
181:         DO 700 ICASE=1,NCASE
182: C..... TO SET UP INITIAL PARAMETER VALUES.
183:         CINIT=C0(ICASE)
184:         C=CINIT
185:         DCDN=0.0
186:         RYCDL=0.0
187:         BCOLD=0.0
188:         CKMOLD=0.0
189:         ICRAK=0
190: C.... TO COMPUTE FOR EACH FLIGHT.
191:         FLIGHT=0.0
192:     720 FLIGHT=FLIGHT+1.0
193:         IF(FLIGHT.GT.1.0) GO TO 725
194:         WRITE(LOUT,100) CINIT
195:         IF(NLOC.GE.0) WRITE(LOUT,105)
196:         IF(NLOC.LT.0) WRITE(LOUT,110)
197:         IF(NTYPE.EQ.0) WRITE(LOUT,115)
198:         IF(NTYPE.NE.0) WRITE(LOUT,120)
199:         IF(NRETRD.LE.0) WRITE(LOUT,125)

```

```

200:      IF(NRETRD.EQ.1)WRITE(LOUT,130)
201:      IF(NRETRD.GT.0.AND.NRETRD.NE.1)WRITE(LOUT,135)
202:      IF(NPLANE.EQ.0)WRITE(LOUT,140)
203:      IF(NPLANE.NE.0)WRITE(LOUT,145)
204:      WRITE(LOUT,150)
205:      100 FORMAT(1H1//1X,86(1H-)/
206:      *11X,'CRACK GROWTH PREDICTION PER FLIGHT --- FOR C0= ',
207:      *F8.4,' (MM)')
208:      105 FORMAT(11X,'----- FOR THROUGH THE THICKNESS CENTER CRACK')
209:      110 FORMAT(11X,'----- FOR THROUGH THE THICKNESS SINGLE EDGE CRACK')
210:      115 FORMAT(11X,'----- PARIS'' GROWTH RATE EQUATION IS APPLIED.')
211:      120 FORMAT(11X,'----- WALKER''S GROWTH RATE EQUATION IS APPLIED.')
212:      125 FORMAT(11X,'----- NO RETARDATION IS CONSIDERED.')
213:      130 FORMAT(11X,'----- WILLENBORG''S RETARDATION MODEL IS USED.')
214:      135 FORMAT(11X,'----- VROMAN''S RETARDATION MODEL IS USED.')
215:      140 FORMAT(11X,'----- PLASTIC ZONE SIZE FOR PLANE STRESS')
216:      145 FORMAT(11X,'----- PLASTIC ZONE SIZE FOR PLANE STRAIN')
217:      150 FORMAT(1X,86(1H-)//
218:      *1X,'FLIGHT NO.',24X,'PREDICTED CRACK GROWTH (MM)'/
219:      *1X,10(1H-),24X,27(1H-))
220:      C.... TO CALCULATE FOR EACH CYCLE IN A FLIGHT.
221:      725 DO 740 ICYC=1,NLOAD,
222:      BIDCYC=0.0
223:      DOCYC=0.0
224:      SIGMAX=SGMAX(ICYC)
225:      SIGMIN=SGMIN(ICYC)
226:      R=SIGMIN/SIGMAX
227:      IF(R.GT.RCUT) R=RCUT
228:      202 CPL=1.005*C
229:      IF(CPL.GE.CTR) GO TO 260
230:      ROOTC=SQRT(PI*CPL)
231:      CALL SIFLOC(NNLOC,SIGMAX,SIGMIN,CKMAX,CKMIN)
232:      IF(CKAMX.GE.CKIC) GO TO 260
233:      DELTAK=CKMAX-CKMIN
234:      CKLIM=COEFK*SIGLIM
235:      IF(CKLIM.GE.CKIC) GO TO 260
236:      C
237:      C.... RETARDATION TREATMENT IN THIS PROGRAM.
238:      C IF NRTRD.LE.0 --- NO RETARDATION IS CONSIDERED.
239:      IF(NRETRD.LE.0) GO TO 210
240:      C.... RETARDATION EFFECT IS TO BE CONSIDERED.
241:      RYC=(CKMAX/SIGMAX)**2/PLANE
242:      BC=CPL+RYC
243:      IF(BC.LE.BCOLD) GO TO 205
244:      RYCOLD=RYC
245:      BCOLD=BC
246:      CKMOLD=CKMAX
247:      C
248:      C..... FOR VROMAN'S RETARDATION MODEL.
249:      205 IF(NRETRD.EQ.1) GO TO 10
250:      CKRES=SQRT((BCOLD-CPL)/RYCOLD)*CKMOLD/3.0
251:      EFFCTK=4.0*(CKMAX-0.75*(CKMIN+CKRES))/3.0
252:      GO TO 15
253:      C
254:      C..... FOR WILLENBORG'S RETARDATION MODEL.
255:      10 CKRES=SQRT((BCOLD-CPL)/RYCOLD)*CKMOLD-CKMAX
256:      ECKMAX=CKMAX-CKRES
257:      ECKMIN=CKMIN-CKRES
258:      C..... FOR MODE SELECTION ACCORDING TO ECKMAX.
259:      IF(ECKMAX) 13,12,11
260:      C FOR MODE-I (RETARDATION DUE TO REDUCED DELTAK AND R).
261:      11 IF(ECKMIN.LT.0.0) ECKMIN=0.0
262:      R=ECKMIN/ECKMAX
263:      IF(R.GT.RCUT) R=RCUT
264:      EFFCTK=ECKMAX-ECKMIN
265:      GO TO 15
266:      C FOR MODE-II (RETARDATION DUE ONLY TO REDUCED R).
267:      12 EFFCTK=DELTAK

```

```

268:          R=0.0
269:          GO TO 15
270: C          FOR MODE-III (RETARDATION MAXIMUM -- GROWTH STOPPED).
271:          13  EFFCTK=0.0
272:          R=0.0
273:          15 DELTAK=EFFCTK
274:          210 RCOMPL=1.0-R
275:             IF (DELTAK.LE.DELKTH) GO TO 740
276:             DCDN=CONST*(RCOMPL**EXPM1*DELTAK)**EXPN
277:             IF (DCDN.GT.0.0) GO TO 218
278:             GO TO 740
279:          218 DC=C/100.0
280:             DN=DC/DCDN
281:             DIDCYC=DIDCYC+DN
282:             IF (DIDCYC-1.0)220,220,230
283:          220 CNEW=C+DC
284:             GO TO 240
285:          230 DIDCYC=DIDCYC-DN
286:             DDCYC=1.0-DIDCYC
287:             CNEW=C+DCDN*DDCYC
288:             DIDCYC=1.0
289:          240 IF (CNEW-CTR) 250,260,260
290:          250 C=CNEW
291:             IF (DIDCYC.GE.1.0) GO TO 740
292:             GO TO 202
293:          740 CONTINUE
294: C..... TO PRINT OUT CRACK SIZE AT EVERY 100 FLIGHTS.
295:          ICRAK=ICRAK+1
296:          CRAK(ICRAK)=C
297:          IF (ICRAK.LT.100) GO TO 720
298:          CALL WTCRK(CRAK,ICRAK,FLIGHT,LOUT)
299: C..... TO COMPUTE FOR NEXT FLIGHT.
300:          GO TO 720
301: C
302: C.... CRACK GROWN TO FAILURE.
303: C.... PRINT OUT THE RESIDUAL CRACK GROWTH DATA.
304:          260 CLAST=C
305:             FCYCLE=(FLIGHT-1.0)*FNLOAD+FLOAT(ICYC)
306:             FTOTAL=FCYCLE/FNLOAD
307:             IF (ICRAK.EQ.0) GO TO 265
308:             CALL WTCRK(CRAK,ICRAK,FLIGHT,LOUT)
309:             GO TO 270
310:          265 IF (FLIGHT.LE.1.0) WRITE (LOUT,267)
311:          267 FORMAT(11X,5(1H-),' FAILURE HAS OCCURRED WITHIN THE FIRST ',
312:             *'FLIGHT ',5(1H-))
313:          270 WRITE(2,775) NLOAD,FCYCLE,FTOTAL,CINIT,CLAST,ICASE
314:          775 FORMAT(1H ,86(1H-)////1H ,60(1H-)/
315:             *1H , 'NO. OF CYCLES IN A FLIGHT  NLOAD= ',I6,9X,' (CYCLES)'/
316:             *1H , 'NO. OF CYCLES TO FAILURE  NCYCLE= ',E15.7,' (CYCLES)'/
317:             *1H , 'NO. OF FLIGHTS TO FAILURE  NFLIGHT= ',E15.7,' (FLIGHTS)'/
318:             *1H , 'INITIAL CRACK SIZE  CO= ',E15.7,' (MM)'/
319:             *1H , 'LAST CRACK LENGTH  CLAST= ',E15.7,' (MM)'/
320:             *1H , 'CASE NUMBER  ICASE= ',I6/1X,60(1H-))
321: C
322: C.... COMPUTE FOR THE NEXT PROBLEM.
323:          700 CONTINUE
324:             GO TO 380
325:          960 WRITE(2,390)
326:          390 FORMAT(1H/1H , 'THE MODEL IS INCORRECT. '//
327:             *1H , 'ANALYSIS IS STOPPED. CHECK INPUT CARDS. '//)
328:          380 STOP
329:             END
330:             SUBROUTINE SIFLOC(NNLOC,SIGMAX,SIGMIN,CKMAX,CKMIN)
331:             COMMON /KLOC/ CPL,HALFW,W,ROOTC,COEFK,PI2
332: C.... SUBRRoutine PROGRAM TO COMPUTE STRESS INTENSITY FACTOR
333: C FOR THROUGH THE THICKNESS CENTER OR SINGLE EDGE CRACK.
334: C IF NNLOC .GE. 0 --- FOR CENTER CRACK
335: C IF NNLOC .LT. 0 --- FOR SINGLE EDGE CRACK

```

```

336:         IF(NNLOC.LT.0) GO TO 10
337:         D=CPL/HALFW
338:         WCORCT=SQRT(1.0/COS(PI2*D))
339:         GO TO 20
340:     10 D=CPL/W
341:         DD=D*D
342:         WCORCT=1.12-0.231*D+10.55*DD-21.72*D*DD+30.39*DD*DD
343:     20 COEFK=ROOCT*WCORCT
344:         CKMAX=COEFK*SIGMAX
345:         CKMIN=COEFK*SIGMIN
346:         RETURN
347:     END
348:     SUBROUTINE WTCRK(X,I,FLIGHT,LOUT)
349:     C.... TO PRINT OUT CRACK GROWTH DATA AT EVERY 100 FLIGHTS.
350:     DIMENSION X(1)
351:     LOGICAL FLAG
352:     K=-4
353:     FLAG=.FALSE.
354:     AF=AMOD(FLIGHT,100.0)
355:     IF(AF.NE.0.0) GO TO 10
356:     FO=FLIGHT-100.0
357:     GO TO 12
358:     10 FO=FLIGHT-AF
359:     12 J1=I/5
360:         J2=I-J1*5
361:         IF(J2.EQ.0) FLAG=.TRUE.
362:         IF(J1.EQ.0) GO TO 100
363:         DO 20 L=1,J1
364:             K=K+5
365:             F=FO+FLOAT(K)
366:             WRITE(LOUT,200) F,X(K),X(K+1),X(K+2),X(K+3),X(K+4)
367:     20 CONTINUE
368:     IF(AF.EQ.0.0) WRITE(LOUT,250)
369:     100 IF(FLAG) GO TO 110
370:         L=J1+1
371:         K=K+5
372:         F=FO+FLOAT(K)
373:         IF(J2.EQ.1) WRITE(LOUT,210) F,X(K)
374:         IF(J2.EQ.2) WRITE(LOUT,220) F,X(K),X(K+1)
375:         IF(J2.EQ.3) WRITE(LOUT,230) F,X(K),X(K+1),X(K+2)
376:         IF(J2.EQ.4) WRITE(LOUT,240) F,X(K),X(K+1),X(K+2),X(K+3)
377:     110 CONTINUE
378:         I=0
379:         200 FORMAT(1X,F10.0,5(1X,E14.7))
380:         210 FORMAT(1X,F10.0,1X,E14.7)
381:         220 FORMAT(1X,F10.0,2(1X,E14.7))
382:         230 FORMAT(1X,F10.0,3(1X,E14.7))
383:         240 FORMAT(1X,F10.0,4(1X,E14.7))
384:         250 FORMAT(1X,5X,80(1H-))
385:         RETURN
386:     END

```

APPENDIX—2 シミュレーション結果の打出し例 (本文中の図7の入力条件  
に対応する結果である)

PREDICTION OF CRACK GROWTH BASED ON WALKER'S EQUATION  
CASE-2: WALKER'S EQUATION; WILLENBORG' MODEL; CENTER CRACK; CO=3.0MM  
FOR THROUGH THE THICKNESS CENTER CRACK  
WILLENBORG'S RETARDATION MODEL IS APPLIED.

MATERIAL CONSTANTS USED:

YIELD STRENGTH SY= .5000000E+02 (KG/MM\*\*2)  
CRITICAL S.I.F. KIC= .6800000E+02 (KG/MM\*\*2)SQRT(MM)  
THRESHOLD S.I.F. KTH= .2500000E+01 (KG/MM\*\*2)SQRT(MM)  
EXPONENT M M= .3100000E+00  
EXPONENT N N= .3400000E+01  
INTERCEPT C C= .1750000E-08  
CUT-OFF RATIO RCUT= .5000000E+00

SPECIMEN GEOMETRY:

HALF WIDTH W/2= .1000000E+02 (MM)  
THICKNESS T= .2000000E+01 (MM)

INPUT SPECTRUM LOADING INFORMATION

DESIGN LIMIT STRESS SIGLIM = .200000E+02 (KG/MM\*\*2)

MAXIMUM STRESS AS PRESCRIBED (KG/MM\*\*2)

4.00	7.60	6.00	7.40	7.40	7.60	5.60	6.20	6.60	7.00
5.80	7.20	5.40	7.60	6.20	5.20	5.40	7.20	6.00	10.60
5.40	7.20	7.80	10.40	4.80	6.80	8.00	4.00	3.80	4.60
9.20	6.20	4.60	3.60	7.80	8.60	7.40	7.60	7.60	5.00
6.40	7.40	11.80	5.80	5.60	8.60	4.80	6.60	7.60	6.60

MINIMUM STRESS AS PRESCRIBED (KG/MM\*\*2)

0.00	1.60	2.80	4.60	2.80	2.00	2.00	1.40	2.80	3.40
1.20	3.00	2.00	3.00	8.40	3.60	2.00	4.00	2.80	.80
2.60	2.80	3.40	1.80	3.00	1.80	4.80	2.00	2.60	1.80
2.60	3.60	2.20	2.20	1.60	5.00	3.00	4.40	4.60	2.80
3.00	1.80	5.80	2.60	2.20	2.80	3.00	1.60	2.20	1.80

MAXIMUM STRESS AS MODIFIED (KG/MM\*\*2)

4.00	7.60	6.00	7.40	7.40	7.60	5.60	6.20	6.60	7.00
5.80	7.20	5.40	7.60	6.20	5.20	5.40	7.20	6.00	10.60
5.40	7.20	7.80	10.40	4.80	6.80	8.00	4.00	3.80	4.60
9.20	6.20	4.60	3.60	7.80	8.60	7.40	7.60	7.60	5.00
6.40	7.40	11.80	5.80	5.60	8.60	4.80	6.60	7.60	6.60

MINIMUM STRESS AS MODIFIED (KG/MM\*\*2)

0.00	1.60	2.80	4.60	2.80	2.00	2.00	1.40	2.80	3.40
1.20	3.00	2.00	3.00	8.40	3.60	2.00	4.00	2.80	.80
2.60	2.80	3.40	1.80	3.00	1.80	4.80	2.00	2.60	1.80
2.60	3.60	2.20	2.20	1.60	5.00	3.00	4.40	4.60	2.80
3.00	1.80	5.80	2.60	2.20	2.80	3.00	1.60	2.20	1.80

PRESCRIBED MEAN STRESS RANGE S1= .43196E+01

MODIFIED MEAN STRESS RANGE S2= .43196E+01

SIMPLE MEAN STRESS-RATIO R1= .4314

RAISED MEAN STRESS-RATIO R2= .4777

結果の打出し例一つづき—

CRACK GROWTH PREDICTION PER FLIGHT --- FOR CO= 3.0000 (MM)  
 ---- FOR THROUGH THE THICKNESS CENTER CRACK  
 ---- WALKER'S GROWTH RATE EQUATION IS APPLIED.  
 ---- WILLENBORG'S RETARDATION MODEL IS USED.  
 ---- PLASTIC ZONE SIZE FOR PLANE STRAIN

FLIGHT NO.	PREDICTED CRACK GROWTH (MM)				
1.	.3001477E+01	.3002271E+01	.3003064E+01	.3003858E+01	.3004653E+01
6.	.3005448E+01	.3006244E+01	.3007040E+01	.3007836E+01	.3008633E+01
11.	.3009431E+01	.3010228E+01	.3011027E+01	.3011826E+01	.3012625E+01
16.	.3013425E+01	.3014225E+01	.3015026E+01	.3015828E+01	.3016629E+01
21.	.3017431E+01	.3018234E+01	.3019038E+01	.3019841E+01	.3020645E+01
26.	.3021449E+01	.3022254E+01	.3023059E+01	.3023865E+01	.3024671E+01
31.	.3025478E+01	.3026285E+01	.3027092E+01	.3027900E+01	.3028708E+01
36.	.3029517E+01	.3030326E+01	.3031136E+01	.3031945E+01	.3032756E+01
41.	.3033567E+01	.3034380E+01	.3035193E+01	.3036006E+01	.3036819E+01
46.	.3037639E+01	.3038448E+01	.3039263E+01	.3040078E+01	.3040894E+01
51.	.3041710E+01	.3042526E+01	.3043343E+01	.3044161E+01	.3044979E+01
56.	.3045798E+01	.3046617E+01	.3047436E+01	.3048256E+01	.3049076E+01
61.	.3049877E+01	.3050719E+01	.3051541E+01	.3052363E+01	.3053186E+01
66.	.3054009E+01	.3054834E+01	.3055659E+01	.3056484E+01	.3057310E+01
71.	.3058136E+01	.3058962E+01	.3059789E+01	.3060616E+01	.3061444E+01
76.	.3062272E+01	.3063100E+01	.3063929E+01	.3064759E+01	.3065589E+01
81.	.3066420E+01	.3067252E+01	.3068083E+01	.3068915E+01	.3069748E+01
86.	.3070581E+01	.3071415E+01	.3072250E+01	.3073084E+01	.3073920E+01
91.	.3074755E+01	.3075592E+01	.3076429E+01	.3077267E+01	.3078105E+01
96.	.3078946E+01	.3079782E+01	.3080622E+01	.3081462E+01	.3082302E+01
101.	.3083143E+01	.3083985E+01	.3084826E+01	.3085669E+01	.3086512E+01
106.	.3087355E+01	.3088199E+01	.3089043E+01	.3089887E+01	.3090732E+01
111.	.3091577E+01	.3092423E+01	.3093270E+01	.3094116E+01	.3094964E+01
116.	.3095812E+01	.3096660E+01	.3097509E+01	.3098359E+01	.3099209E+01
121.	.3100060E+01	.3100911E+01	.3101763E+01	.3102615E+01	.3103468E+01
126.	.3104321E+01	.3105175E+01	.3106031E+01	.3106886E+01	.3107742E+01
131.	.3108599E+01	.3109456E+01	.3110313E+01	.3111171E+01	.3112030E+01
136.	.3112889E+01	.3113748E+01	.3114608E+01	.3115468E+01	.3116329E+01
141.	.3117190E+01	.3118052E+01	.3118914E+01	.3119777E+01	.3120640E+01
146.	.3121504E+01	.3122369E+01	.3123233E+01	.3124098E+01	.3124963E+01
151.	.3125829E+01	.3126696E+01	.3127563E+01	.3128431E+01	.3129298E+01
156.	.3130167E+01	.3131036E+01	.3131905E+01	.3132775E+01	.3133645E+01
161.	.3134516E+01	.3135388E+01	.3136260E+01	.3137132E+01	.3138006E+01
166.	.3138879E+01	.3139754E+01	.3140628E+01	.3141503E+01	.3142379E+01
171.	.3143255E+01	.3144131E+01	.3145008E+01	.3145885E+01	.3146764E+01
176.	.3147642E+01	.3148522E+01	.3149402E+01	.3150282E+01	.3151163E+01
181.	.3152045E+01	.3152927E+01	.3153811E+01	.3154694E+01	.3155578E+01
186.	.3156463E+01	.3157348E+01	.3158234E+01	.3159120E+01	.3160007E+01
191.	.3160894E+01	.3161782E+01	.3162671E+01	.3163560E+01	.3164450E+01
196.	.3165340E+01	.3166232E+01	.3167124E+01	.3168016E+01	.3168910E+01
201.	.3169803E+01	.3170697E+01	.3171592E+01	.3172487E+01	.3173382E+01
206.	.3174278E+01	.3175174E+01	.3176070E+01	.3176968E+01	.3177866E+01
211.	.3178764E+01	.3179663E+01	.3180562E+01	.3181462E+01	.3182363E+01
216.	.3183264E+01	.3184165E+01	.3185067E+01	.3185970E+01	.3186873E+01
221.	.3187777E+01	.3188681E+01	.3189585E+01	.3190491E+01	.3191397E+01
226.	.3192303E+01	.3193211E+01	.3194119E+01	.3195028E+01	.3195937E+01
231.	.3196846E+01	.3197757E+01	.3198668E+01	.3199580E+01	.3200493E+01
236.	.3201406E+01	.3202319E+01			

NO. OF CYCLES IN A FLIGHT NLOAD= 50 (CYCLES)  
 NO. OF CYCLES TO FAILURE NNCYCLE= .1189400E+05 (CYCLES)  
 NO. OF FLIGHTS TO FAILURE NFLIGHT= .2378800E+03 (FLIGHTS)  
 INITIAL CRACK SIZE CO= .3000000E+01 (MM)  
 LAST CRACK LENGTH CLAST= .3203188E+01 (MM)  
 CASE NUMBER ICASE= 1